

## Example 5a: General Linear Models with Multiple Fixed Effects of Multiple Predictors Simultaneously in SAS and STATA (with GSS example data)

The data for this example were selected from the 2012 General Social Survey dataset featured in Mitchell (2015); these data were also used for examples 1, 2, 3, and 4. Building on the results of Example 4 (summarized below), this example will examine the unique effects of three-category working class, quadratic years of age, and piecewise years of education in predicting annual income. It will demonstrate how the results from hierarchical (stepwise) regression can be obtained from a single model using multivariate Wald F-tests.

### SAS Syntax for Importing and Preparing Data for Analysis:

```
* Paste in the folder address where "GSS_Example.xlsx" is saved after = before ;
%LET filesave= \\Client\C:\Dropbox\21SP_PSQF6242\PSQF6242_Example5a;

* IMPORT GSS_Example.xlsx data using filesave reference and exact file name;
* from the Excel workbook in DATAFILE= location from SHEET= ;
* New SAS file is in "work" library place with name "Example5a";
* "GETNAMES" reads in the first row as variable names;
* DBMS=XLSX (can also use EXCEL or XLS for .xls files);
PROC IMPORT DATAFILE="%filesave.\GSS_Example.xlsx"
            OUT=work.Example5a DBMS=XLSX REPLACE;
            SHEET="GSS_Example";
            GETNAMES=YES;
RUN;

* All data transformations must happen inside a DATA+SET combo to know where to use them;
DATA work.Example5a; SET work.Example5a;
* Create and label predictor variables for analysis (same as in Example 4);
* Binary predictors for workclass;
  LvsM=.; LvsU=.; * Make new empty variables;
  IF workclass=1 THEN DO; LvsM=0; LvsU=0; END; * Replace each for lower;
  IF workclass=2 THEN DO; LvsM=1; LvsU=0; END; * Replace each for middle;
  IF workclass=3 THEN DO; LvsM=0; LvsU=1; END; * Replace each for upper;
  LABEL LvsM="LvsM: Low=0 vs Mid=1 Class"
        LvsU="LvsU: Low=0 vs Upp=1 Class";
* Center age at 18;
  age18=age-18;
  LABEL age18= "age18: Age (0=18 years)";
* Piecewise slopes for education;
  lessHS=.; gradHS=.; overHS=.; * Make three new empty variables;
* Replace for educ less than 12;
  IF educ LT 12 THEN DO; lessHS=educ-11; gradHS=0; overHS=0; END;
* Replace for educ greater or equal to 12;
  IF educ GE 12 THEN DO; lessHS=0; gradHS=1; overHS=educ-12; END;
  LABEL lessHS= "lessHS: Slope for Years Ed Less Than High School"
        gradHS= "gradHS: Bump for Graduating High School"
        overHS= "overHS: Slope for Years Ed After High School";
* Label outcome;
  LABEL income= "income: Annual Income in 1000s";
* Select cases complete on variables;
  WHERE NMISS(income,workclass,age,educ)=0;
RUN;
* Now dataset work.Example5a is ready to use;
```

### STATA Syntax for Importing and Preparing Data for Analysis:

```
// Paste in the folder address where "GSS_Example.xlsx" is saved between " "
global filesave "\\Client\C:\Dropbox\21SP_PSQF6242\PSQF6242_Example5a"

// IMPORT GSS_Example.xlsx data using filesave reference and exact file name
// To change all variable names to lowercase, remove "case(preserve)"
clear // Clear before means close any open data
```

```

import excel "$filesave\GSS_Example.xlsx", case(preserve) firstrow clear
// Clear after means re-import if it already exists (if need to start over)

// Create and label predictor variables for analysis (same as in Example 4)
// Binary predictors for workclass
gen LvsM=. // Make two new empty variables
gen LvsU=.
replace LvsM=0 if workclass==1 // Replace each for lower
replace LvsU=0 if workclass==1
replace LvsM=1 if workclass==2 // Replace each for middle
replace LvsU=0 if workclass==2
replace LvsM=0 if workclass==3 // Replace each for upper
replace LvsU=1 if workclass==3
label variable LvsM "LvsM: Low=0 vs Mid=1 Class"
label variable LvsU "LvsU: Low=0 vs Upp=1 Class"
// Center age at 18
gen age18 = age-18
label variable age18 "age18: Age (0=18 years)"
// Piecewise slopes for education
gen lessHS=. // Make 3 new empty variables
gen gradHS=.
gen overHS=.
// Replace for educ less than 12
replace lessHS=educ-11 if educ < 12
replace gradHS=0 if educ < 12
replace overHS=0 if educ < 12
// Replace for educ greater or equal to 12
replace lessHS=0 if educ >= 12
replace gradHS=1 if educ >= 12
replace overHS=educ-12 if educ >= 12
// Label variables
label variable lessHS "lessHS: Slope for Years Ed Less Than High School"
label variable gradHS "gradHS: Acute Bump for Graduating High School"
label variable overHS "overHS: Slope for Years Ed After High School"
// Label outcome
label variable income "income: Annual Income in 1000s"
// Select cases complete on variables of interest
egen nmiss = rowmiss(income workclass age educ)
drop if nmiss>0
// Now dataset is ready to use

```

### Combined Model Predicting Income from 7 Slopes: Three-Category WorkClass, a Quadratic Trends of Age, and Piecewise Linear Slopes for Education:

$$\begin{aligned}
 \text{Income}_i = & \beta_0 + \beta_1(LvsM_i) + \beta_2(LvsU_i) + \beta_3(\text{Age}_i - 18) + \beta_4(\text{Age}_i - 18)^2 \\
 & + \beta_5(LessHS_i) + \beta_6(GradHS_i) + \beta_7(OverHS_i) + e_i
 \end{aligned}$$

In addition to the overall  $F$ -test of the model  $R^2$ , the purpose of estimating a single model with the 7 slopes from all three predictive constructs combined (workclass, age, and education) is to determine to what extent their bivariate effects (when each construct was in a separate model predicting income, as was the case in Example 4) differ from their unique effects (when all constructs are combined in the same model, below). The solution for the fixed effects will provide tests for the significance of each slope (against a null hypothesis of a 0 slope in the population), and we can also ask for joint tests (and their effect sizes) that combine the multiple slopes needed to capture the full effect of each construct.

### SAS Syntax and Output:

```

TITLE "SAS GLM Predicting Income from WorkClass, Age, and Education";
PROC GLM DATA=work.Example5a NAMELEN=100 PLOTS (UNPACK)=DIAGNOSTICS;
* Combined model with all 7 slopes;
MODEL income = LvsM LvsU age18 age18*age18 lessHS gradHS overHS
      / SOLUTION ALPHA=.05 CLPARM SS3 EFFECTSIZE;
* Ask for missing model-implied group difference;
ESTIMATE "Mid vs Upp Diff" LvsM -1 LvsU 1;

* Replicate F-test and R2 for the model: includes all 7 slopes;
CONTRAST "F-test (DFnum=7) for model"
      LvsM 1, LvsU 1, age18 1, age18*age18 1, lessHS 1, gradHS 1, overHS 1;
* Ask for F-test and semi-partial R2 for overall effect of workclass;
CONTRAST "F-test (DFnum=2) for overall workclass" LvsM 1, LvsU 1;
* Ask for F-test and semi-partial R2 for overall effect of age;
CONTRAST "F-test (DFnum=2) for overall age" age18 1, age18*age18 1;
* Ask for F-test and semi-partial R2 for overall effect of education;
CONTRAST "F-test (DFnum=3) for overall education" lessHS 1, gradHS 1, overHS 1;

* Save for computing slope-specific effect sizes;
ODS OUTPUT ParameterEstimates=ModelEstimates Estimates=ReqEstimates;
RUN; QUIT; TITLE;

```

### SAS GLM Predicting Income from Age, Education, and WorkClass

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	7	40246.4243	5749.4892	<b>42.09</b>	<.0001
Error	<b>726</b>	99176.8076	<b>136.6072</b>		
Corrected Total	733	139423.2319			

R-Square	Coeff Var	Root MSE	income Mean
<b>0.288664</b>	67.54893	11.68791	17.30287

**Mean Square Error**, the residual variance, is 136.61 after including 7 slopes for the 3 predictor constructs (which accounted for 28.87% of the variance in income as the model R<sup>2</sup>). The *F*-test says this R<sup>2</sup> is significantly > 0,  $F(7, 726) = 42.04$ ,  $MSE = 136.61$ ,  $p < .001$ .

Source	DF	Type III SS	Mean Square	F Value	Pr > F
LvsM	1	5594.06856	5594.06856	40.95	<.0001
LvsU	1	975.26540	975.26540	7.14	0.0077
age18	1	10336.70737	10336.70737	75.67	<.0001
age18*age18	1	8089.56445	8089.56445	59.22	<.0001
lessHS	1	29.07769	29.07769	0.21	0.6447
gradHS	1	440.94624	440.94624	3.23	0.0728
overHS	1	7370.86851	7370.86851	53.96	<.0001

Because the workclass predictors are related (each shares a reference group with another), the total of the SS values for these three differences they imply (two of which are given here) is greater than it should be. The same is true for the two age predictors, as well as for the three education predictors.

For this reason, the slope-specific  $sr^2$  values are not valid. Instead, we need to obtain an *F*-test and semipartial eta-square effect size that combines the slopes for the same construct... that's what the CONTRAST statements were for! Let's see what they give us...

Source	Total Variation Accounted For			
	Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits	
LvsM	0.0401	0.0391	0.0169	0.0714
LvsU	0.0070	0.0060	0.0000	0.0238
age18	0.0741	0.0731	0.0417	0.1126
age18*age18	0.0580	0.0570	0.0295	0.0935
lessHS	0.0002	-0.0008	0.0000	0.0071
gradHS	0.0032	0.0022	0.0000	0.0163
overHS	0.0529	0.0518	0.0257	0.0873

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
F-test (DFnum=7) for model	7	40246.42433	5749.48919	42.09	<.0001
F-test (DFnum=2) for overall workclass	2	5961.75558	2980.87779	21.82	<.0001
F-test (DFnum=2) for overall age	2	11223.60775	5611.80388	41.08	<.0001
F-test (DFnum=3) for overall education	3	11251.78823	3750.59608	27.46	<.0001

Total Variation Accounted For (SR = SS contrast / SS total)

Contrast	Semipartial		Conservative	
	Eta-Square	Omega-Square	95% Confidence Limits	
F-test (DFnum=7) for model	0.2887 = R <sup>2</sup>	0.2815	0.2303	0.3331
F-test (DFnum=2) for overall workclass	0.0428	0.0408	0.0175	0.0733
F-test (DFnum=2) for overall age	0.0805	0.0785	0.0456	0.1187
F-test (DFnum=3) for overall education	0.0807	0.0777	0.0446	0.1177

The semipartial eta-squares above give the amount of variance accounted for each set of slopes (the sets we requested using CONTRAST statements). Whether those semipartial eta-squared values are > 0 is tested by the corresponding F-value (in prior table). Btw, semipartial omega-square column is analogous to the adjusted R<sup>2</sup>.

**Table from ESTIMATE statement (for model-implied fixed effects)**

Parameter	Estimate	Standard Error	t Value	Pr >  t	95% Confidence Limits	
Mid vs Upp Diff	1.14843248	2.70813034	0.42	0.6716	-4.16826904	6.46513400

**Table of Model-Estimated Fixed Effects (normally is last)**

Parameter	Estimate	Standard Error	t Value	Pr >  t	95% Confidence Limits		
Intercept	-3.686546177	2.00461546	-1.84	0.0663	-7.622081294	0.248988941	<b>Beta0</b>
LvsM	6.060105402	0.94700667	6.40	<.0001	4.200906929	7.919303874	<b>Beta1</b>
LvsU	7.208537879	2.69787938	2.67	0.0077	1.911961423	12.505114336	<b>Beta2</b>
age18	1.069979988	0.12300458	8.70	<.0001	0.828492845	1.311467130	<b>Beta3</b>
age18*age18	-0.017506167	0.00227492	-7.70	<.0001	-0.021972365	-0.013039969	<b>Beta4</b>
lessHS	0.258917912	0.56120164	0.46	0.6447	-0.842853869	1.360689693	<b>Beta5</b>
gradHS	3.157139208	1.75726664	1.80	0.0728	-0.292791564	6.607069980	<b>Beta6</b>
overHS	1.528179214	0.20804233	7.35	<.0001	1.119742828	1.936615600	<b>Beta7</b>

```
* Compute Cohen d effect size and r effect sizes;
DATA work.FinalEstimates; LENGTH Parameter $50;
SET work.ModelEstimates ReqEstimates; %LET df=726;
CohenD=(2*tvalue)/SQRT(&df.);
EffectR=tvalue/SQRT((tvalue*tvalue)+&df.); RUN;
```

```
* Print effect sizes;
TITLE "Effect Sizes for Final Model";
PROC PRINT NOOBS DATA=work.FinalEstimates;
* Do not print intercept;
WHERE INDEX(Parameter,"Intercept")=0;
VAR Parameter Estimate StdErr probt CohenD EffectR; RUN; TITLE;
```

**Effect Sizes for Final Model**

Parameter	Estimate	StdErr	tValue	Probt	CohenD	EffectR
LvsM	6.060105402	0.94700667	6.40	<.0001	0.47499	0.23107
LvsU	7.208537879	2.69787938	2.67	0.0077	0.19833	0.09868
Mid vs Upp Diff	1.148432478	2.70813034	0.42	0.6716	0.03148	0.01574
age18	1.069979988	0.12300458	8.70	<.0001	0.64568	0.30723
age18*age18	-0.017506167	0.00227492	-7.70	<.0001	-0.57120	-0.27462
lessHS	0.258917912	0.56120164	0.46	0.6447	0.03425	0.01712
gradHS	3.157139208	1.75726664	1.80	0.0728	0.13336	0.06653
overHS	1.528179214	0.20804233	7.35	<.0001	0.54524	0.26302

## STATA Syntax and Output:

```

display "STATA GLM Predicting Income from WorkClass, Age, and Education"
regress income c.LvsM c.LvsU c.age18 c.age18#c.age18 c.lessHS c.gradHS c.overHS, level(95)
// Ask for missing model-implied group difference
lincom c.LvsM*-1 + c.LvsU*1 // Mid vs Upp Diff

// Replicate F-test for the model: includes all 7 slopes
test (c.LvsM=0) (c.LvsU=0) (c.age18=0) (c.age18#c.age18=0) ///
(c.lessHS=0) (c.gradHS=0) (c.overHS=0)
// Ask for F-test for overall effect of workclass
test (c.LvsM=0) (c.LvsU=0)
// Ask for F-test for overall effect of age
test (c.age18=0) (c.age18#c.age18=0)
// Ask for F-test for overall effect of education
test (c.lessHS=0) (c.gradHS=0) (c.overHS=0)

```

## STATA GLM Predicting Income from WorkClass, Age, and Education

Source	SS	df	MS	Number of obs	=	734
Model	40246.4243	7	5749.48919	F(7, 726)	=	42.09
Residual	99176.8076	726	136.607173	Prob > F	=	0.0000
				R-squared	=	0.2887
				Adj R-squared	=	0.2818
Total	139423.232	733	190.209048	Root MSE	=	11.688

income	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
LvsM	6.060105	.9470067	6.40	0.000	4.200907 7.919304
LvsU	7.208538	2.697879	2.67	0.008	1.911961 12.50511
age18	1.06998	.1230046	8.70	0.000	.8284928 1.311467
c.age18#c.age18	-.0175062	.0022749	-7.70	0.000	-.0219724 -.01304
lessHS	.2589179	.5612016	0.46	0.645	-.8428539 1.36069
gradHS	3.157139	1.757267	1.80	0.073	-.2927916 6.60707
overHS	1.528179	.2080423	7.35	0.000	1.119743 1.936616
_cons	-3.686546	2.004615	-1.84	0.066	-7.622081 .2489889

```

. // Ask for model-implied group difference
. lincom c.LvsM*-1 + c.LvsU*1 // Mid vs Upp Diff

```

( 1) - LvsM + LvsU = 0

income	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]
(1)	1.148432	2.70813	0.42	0.672	-4.168269 6.465134

```

. // Replicate F-test for the model: includes all 7 slopes
. test (c.LvsM=0) (c.LvsU=0) (c.age18=0) (c.age18#c.age18=0) ///
> (c.lessHS=0) (c.gradHS=0) (c.overHS=0)

```

```

( 1) LvsM = 0
( 2) LvsU = 0
( 3) age18 = 0
( 4) c.age18#c.age18 = 0
( 5) lessHS = 0
( 6) gradHS = 0
( 7) overHS = 0

```

```

F( 7, 726) = 42.09
Prob > F = 0.0000

```

```
. // Ask for F-test for overall effect of workclass
. test (c.LvsM=0) (c.LvsU=0)
```

```
( 1) LvsM = 0
( 2) LvsU = 0
```

```
F( 2, 726) = 21.82
Prob > F = 0.0000
```

```
. // Ask for F-test for overall effect of age
. test (c.age18=0) (c.age18#c.age18=0)
```

```
( 1) age18 = 0
( 2) c.age18#c.age18 = 0
```

```
F( 2, 726) = 41.08
Prob > F = 0.0000
```

```
. // Ask for F-test for overall effect of education
. test (c.lessHS=0) (c.gradHS=0) (c.overHS=0)
```

```
( 1) lessHS = 0
( 2) gradHS = 0
( 3) overHS = 0
```

```
F( 3, 726) = 27.46
Prob > F = 0.0000
```

```
display as result "STATA Reduced Model to Get SS for workclass (not included)"
regress income c.age18 c.age18#c.age18 c.lessHS c.gradHS c.overHS, level(95)
// sr2 = (SSfull-SSreduced)/SStotal
display (40246.42433-34284.6688)/139423.2319 // sr2 for workclass = .04276013
```

Source	SS	df	MS	Number of obs	=	734
Model	34284.6688	5	6856.93375	F(5, 728)	=	47.48
Residual	105138.563	728	144.421103	Prob > F	=	0.0000
Total	139423.232	733	190.209048	R-squared	=	0.2459
				Adj R-squared	=	0.2407
				Root MSE	=	12.018

```
display "STATA Reduced Model to Get SS for age (not included)"
regress income c.LvsM c.LvsU c.lessHS c.gradHS c.overHS, level(95)
// sr2 = (SSfull-SSreduced)/SStotal
display (40246.42433-29022.8166)/139423.2319 // sr2 for age = .08050027
```

Source	SS	df	MS	Number of obs	=	734
Model	29022.8166	5	5804.56332	F(5, 728)	=	38.28
Residual	110400.415	728	151.648922	Prob > F	=	0.0000
Total	139423.232	733	190.209048	R-squared	=	0.2082
				Adj R-squared	=	0.2027
				Root MSE	=	12.315

```
display "STATA Reduced Model to Get SS for education (not included)"
regress income c.LvsM c.LvsU c.age18 c.age18#c.age18, level(95)
// sr2 = (SSfull-SSreduced)/SStotal
display (40246.42433-28994.6361)/139423.2319 // sr2 for education = .08070239
```

Source	SS	df	MS	Number of obs	=	734
Model	28994.6361	4	7248.65903	F(4, 729)	=	47.85
Residual	110428.596	729	151.479555	Prob > F	=	0.0000
Total	139423.232	733	190.209048	R-squared	=	0.2080
				Adj R-squared	=	0.2036
				Root MSE	=	12.308

Here is a comparison of the results from each construct in a separate model (from Example 4) with the present results from a combined model with all 7 slopes

Separate Models						
	DF num	Effect SS	Total SS	Model R2	SR2	
2 Group Differences Workclass	2	14414	139423	0.103	0.103	
Linear + Quadratic Age	2	15885	139423	0.114	0.114	
3 Piecewise Slopes Education	3	22907	139423	0.164	0.164	
Combined Model						
	DF num	Effect SS	Total SS	Model R2	SR2	change in SR2
Full Model	7	40246	139423	0.289	0.289	
2 Group Differences Workclass	2	5962	139423	0.289	0.043	0.061
Linear + Quadratic Age	2	11224	139423	0.289	0.081	0.033
3 Piecewise Slopes Education	3	11252	139423	0.289	0.081	0.084
Sum of Workclass, Age, Education	7	28437	418270		0.204	

Above: The sum across the three constructs of the Effects Sums of Squares (SS) is less than the Model SS—this is because the Model SS takes into account the predictive contribution of the shared variance among the sets of predictors (i.e., none of them “gets credit” for what they have in common that predicts income, but the model R<sup>2</sup> does reflect that common contribution). Below: Bivariate slopes (separate) versus unique slopes (combined)

Effect from Construct-Separate Models	Est	SE	t	p	DF den	d	r
Lower vs Middle Class	8.854	1.004	8.822	<.0001	731	0.65	0.310
Lower vs Upper Class	10.985	2.990	3.673	0.000	731	0.27	0.135
(Middle vs Upper Class)	2.130	3.027	0.704	0.482	731	0.05	0.026
Linear Age Slope	1.223	0.135	9.055	<.0001	731	0.67	0.318
Quadratic Age Slope	-0.020	0.003	-7.809	<.0001	731	-0.58	-0.277
Education 2 to 11 years	-0.269	0.599	-0.449	0.654	730	-0.03	-0.017
Education: 11 to 12 years	4.685	1.876	2.498	0.013	730	0.18	0.092
Education: 12 to 20 years	2.125	0.214	9.941	<.0001	730	0.74	0.345
Effect from Combined Model	Est	SE	t	p	DF den	d	r
Lower vs Middle Class	6.060	0.947	6.400	<.0001	726	0.47	0.231
Lower vs Upper Class	7.209	2.698	2.670	0.008	726	0.20	0.099
(Middle vs Upper Class)	1.148	2.708	0.420	0.672	726	0.03	0.016
Linear Age Slope	1.070	0.123	8.700	<.0001	726	0.65	0.307
Quadratic Age Slope	-0.018	0.002	-7.700	<.0001	726	-0.57	-0.275
Education 2 to 11 years	0.259	0.561	0.460	0.645	726	0.03	0.017
Education: 11 to 12 years	3.157	1.757	1.800	0.073	726	0.13	0.067
Education: 12 to 20 years	1.528	0.208	7.350	<.0001	726	0.55	0.263
Difference: Separate Minus Combined	Est					d	r
Lower vs Middle Class	2.794					0.178	0.079
Lower vs Upper Class	3.776					0.073	0.036
(Middle vs Upper Class)	0.982					0.021	0.010
Linear Age Slope	0.153					0.024	0.010
Quadratic Age Slope	-0.002					-0.006	-0.003
Education 2 to 11 years	-0.528					-0.067	-0.034
Education: 11 to 12 years	1.528					0.052	0.026
Education: 12 to 20 years	0.596					0.191	0.082

**Example Results Section (would continue from separate results described in Example 4):**

[Table 1 would report the parameter estimates from the combined model, along with d and r effect sizes. The table note would indicate how they were computed:  $d = \frac{2t}{\sqrt{DF_{den}}}$ ;  $r = \frac{t}{\sqrt{t^2 + DF_{den}}}$ ]

After examining the bivariate contributions of three-category self-reported working class membership, linear and quadratic years of age, and piecewise slopes for years of education in separate models, we then estimated a combined model to examine their unique contributions after controlling for each other predictor variable. The model including all seven slopes captured a significant amount of variance in annual income,  $F(7, 726) = 42.09$ ,  $MSE = 136.61$ ,  $p < .001$ ,  $R^2 = .289$ . Parameter estimates and effect sizes are given in Table 1. Semipartial eta-squared ( $\eta^2$ ) effect sizes and corresponding multivariate Wald F-tests were obtained to evaluate the amount of variance captured by distinct sets of predictor slopes.

The omnibus unique effect of three-category self-reported working class membership remained significant,  $F(2, 726) = 21.83$ ,  $MSE = 136.61$ ,  $p < .0001$ , semipartial  $\eta^2 = .043$ . As shown in Table 1, relative to lower-class respondents (the reference group), after controlling for years of age and years of education, annual income was still significantly higher for both middle-class and upper-class respondents (by 6.060 and 7.209 thousand dollars, respectively). Middle-class and upper-class respondents still did not differ significantly in predicted annual income.

The omnibus unique effect of quadratic years of age (centered at 18) also remained significant,  $F(2, 726) = 41.08$ ,  $MSE = 136.61$ ,  $p < .0001$ , semipartial  $\eta^2 = .081$ . As shown in Table 1, after controlling for self-reported working class and years of education, annual income was expected to be significantly higher by 1.070 thousand dollars per year of age at age 18; this instantaneous linear age slope was predicted to become significantly less positive per year of age by twice the quadratic coefficient of  $-0.018$ . As given by the quantity  $(-1 * \text{linear slope}) / (2 * \text{quadratic slope}) + 18$ , the age of maximum predicted personal income was 48.56 (i.e., the age at which the linear age slope = 0).

The omnibus unique effect of piecewise years of education (centered at 11) also remained significant,  $F(3, 726) = 27.46$ ,  $MSE = 136.61$ ,  $p < .0001$ , semipartial  $\eta^2 = .081$ . As shown in Table 1, after controlling for self-reported working class and years of age, annual income was expected to be nonsignificantly higher by 0.259 thousand dollars per year of education from 2 to 11 years, to be nonsignificantly higher by 3.157 thousand dollars for those achieving a high school degree, and to be significantly higher by 1.528 thousand dollars per year of additional education past 12 years. Notably, the effect of a high school degree (the difference between 11 and 12 years of education) was no longer significant after controlling for age and self-reported working class membership.

[Emphasize why it matters based on your research questions that the predictors had significant unique effects.]